

## Datos para Manejo de Archivos

DATA **países**;  
INFILE CARDS TRUNCOVER;  
INPUT pais \$10. continente \$10.;  
CARDS;  
Chile América  
Ecuador América  
Canadá América  
España Europa  
China Asia  
Alemania Europa  
Run;

DATA **capitales**;  
INFILE CARDS TRUNCOVER;  
INPUT pais \$10. capital \$10. pres \$10.;  
CARDS;  
Chile Santiago Lagos  
Ecuador Quito Gutiérrez  
Canadá Ottawa Clarkson  
España Madrid Asnar  
China Beijing Jiang  
Francia Paris Chirac  
RUN;

DATA **facts**;  
INFILE CARDS TRUNCOVER;  
INPUT export \$10. poprate 3.2  
m\_f\_rate 3.2 continente \$10. pais \$10.;  
CARDS;  
frutas 109105América Chile  
textiles 026103Europa Alemania  
camarones 196098América Ecuador  
maquinaria009106Europa España  
ron 051092América P. Rico  
RUN;

**trans.txt**  
(variables: cuenta, nombre, dirección, transacción)

25-847	100	
22-255	50	
25-847	María Heros	50
22-255	200	
25-847	80	
65-214	19	Fisher

**Banco.txt**  
(variables: cuenta, nombre, dirección, transacción)

22-255Jose Saturnino 55 Newton 1220.33  
25-847Marta Heros 62 Kepler 985321.75  
63-254Pedro Perez 54 Pasteur66481.24  
65-214Juan Sinmiedo 23 Chile 652.02

**Curso.txt**  
(variables: nombre, ex1-ex4, sección)

Alvarez, Maria	87 89 78 85 01
Baez, Pedro	95 96 90 91 02
Castro, Marta	97 98 96 94 02
Garcia, Juan	75 74 72 70 01
Martinez, Roberto	65 62 68 60 02
Perez, Sandra	75 78 72 74 01
Vera, Lisa	85 87 86 82 01

**Horario.txt**  
(variables: día, clase, sección)

Lunes ESTA 3045 001 ESTA 3042 004 ESTA 3041 005  
Martes ESTA 3045 001 ESTA 3042 007 ESTA 4206 0U1  
Miércoles ESTA 3045 001 ESTA 3042 004 ESTA 3041 005  
Jueves ESTA 3045 001 ESTA 3042 007

Dra. Ada Ray

# Programas para Manejo de Archivos

## Stacking e Intercalando

### Ejemplo 1

```
LIBNAME datos 'c:\e4206'; RUN;  
  
DATA stack;  
  SET datos.países datos.capitales (DROP=pres);  
PROC PRINT;  
RUN;
```

### Ejemplo 2

```
PROC SORT DATA=datos.países;  
  BY país;  
PROC SORT DATA=datos.capitales;  
  BY país;  
  
DATA stack2;  
  SET datos.países datos.capitales (DROP=pres);  
  BY país;  
PROC PRINT;  
RUN;
```

### Ejemplo 3 - resultados no deseados

```
DATA stack3;  
  SET datos.países;  
  SET datos.capitales;  
PROC PRINT;  
RUN;
```

## Merging

### Ejemplo 1 - resultados no deseados cuando no se pone el BY

```
DATA comb;  
  MERGE datos.países datos.capitales (DROP=pres);  
  * BY país;  
  
PROC PRINT;  
RUN;
```

## Ejemplo 2

```
PROC SORT DATA=datos.países;  
BY continente país;  
PROC SORT DATA=datos.facts;  
BY continente país;  
DATA comb;  
MERGE datos.países datos.facts;  
BY continente país;  
PROC PRINT;  
RUN;
```

## Ejemplo 3 -resultados no deseados

```
PROC SORT DATA=datos.países;  
BY continente;  
PROC SORT DATA=datos.facts;  
BY continente;  
DATA comb;  
MERGE datos.países datos.facts;  
BY continente;  
PROC PRINT;  
RUN;
```

## Combinando estadísticos con data set original

```
DATA banco;  
INPUT cuenta $ trans @@;  
CARDS;  
1-23 10 2-54 5 3-62 6 1-23 3 2-54 7  
RUN;  
PROC SORT DATA= banco OUT=bank;  
BY cuenta;  
  
/*Guarda total de transacciones por cuenta*/  
PROC MEANS NOPRINT DATA=bank;  
VAR trans;  
BY cuenta;  
OUTPUT OUT=resumen SUM(trans)=total;  
PROC PRINT DATA=resumen;  
TITLE 'Resumen de los datos';  
DATA conjunto;  
MERGE bank resumen;  
BY cuenta;  
DROP _type_ _freq_;  
percent=trans/total*100;  
PROC PRINT DATA=conjunto NOOBS;
```

```
BY cuenta;
ID cuenta;
VAR trans total percent;
TITLE2 'Por cientos';
RUN;
```

## **Combinando un total con data set original**

```
DATA banco;
INPUT cuenta $ trans @@;
CARDS;
1-23 10 2-54 5 3-62 6 1-23 3 2-54 7
RUN;
```

```
/*Guarda total de transacciones*/
PROC MEANS NOPRINT DATA=banco;
VAR trans;
OUTPUT OUT=resumen SUM(trans)=total;
```

```
PROC PRINT DATA=resumen;
TITLE 'Resumen de los datos';
```

```
/*Combina los dos archivos: resumen y banco*/
DATA conjunto;
IF _N_=1 THEN SET resumen;
SET banco;
percent=trans/total*100;
PROC SORT;
BY cuenta;
PROC PRINT DATA=conjunto NOOBS;
VAR cuenta trans total percent;
BY cuenta;
TITLE2 'Por cientos';
RUN;
```

## **Actualización de un archive de datos**

```
DATA transacciones;
INFILE 'c:\temp\trans.txt' MISSOVER;
INPUT cuenta $6. nombre $16. dir $10.trans;
PROC SORT;
BY cuenta;
```

```
/*Actualiza data set banco*/
DATA datos.banco;
UPDATE datos.banco transacciones;
```

```

    BY cuenta;

PROC MEANS NOPRINT DATA=transacciones;
VAR trans;
BY cuenta;
OUTPUT OUT=balances SUM(trans)=trans;

/*Actualiza el archive master Banco y calcula el nuevo balance*/
DATA datos.banco;
UPDATE datos.banco balances;
BY cuenta;
IF trans=. THEN trans=0;
balance=balance-trans;
DROP _type _freq _trans;
PROC PRINT DATA=datos.banco;
RUN;

```

## OUTPUT

### Ejemplo 1 – dividir un archivo en varios

```

DATA seccion_1 seccion_2;
INFILE 'c:\temp\curso.txt';
INPUT nombre $25. ex1-ex4 sec $;
IF sec='01' THEN OUTPUT seccion_1;
    ELSE OUTPUT seccion_2;
PROC PRINT DATA =seccion_1;
TITLE 'Sección 1';
PROC PRINT DATA =seccion_2;
TITLE 'Sección 2';
RUN;

```

### Ejemplo 2 – crear varias observaciones a partir de una

```

DATA programa;
INFILE 'c:\temp\horario.txt';
INPUT dia $10. clase $10. sec @;
OUTPUT;
INPUT clase $10. sec @;
OUTPUT;
INPUT clase $10. sec;
OUTPUT;
PROC PRINT DATA=programa NOOBS;
TITLE 'Programa de clase';
RUN;

```